Key Homework 21

Math 277, Fall 2005

Ordinary Differential Equations

Dr. Goutziers

Initializations

```
> restart;
```

1: Page 219

Code the differential equation, find the auxiliary equation, compute the eigenvalues.

```
> deq:=3*diff(y(t), t$2)+48*y(t)=0;
  aux:=simplify(eval(subs(y(t)=exp(r*t), deq))/exp(r*t));
  evals:=solve(aux, r);
```

$$deq := 3\left(\frac{d^2}{dt^2}\mathrm{y}(t)\right) + 48\,\mathrm{y}(t) = 0$$

$$aux := 3\,r^2 + 48 = 0$$

$$evals := 4\,I, -4\,I$$

Find the general solution and implement the initial conditions.

```
> Y[1]:=cos(Im(evals[1])*t);
  Y[2]:=sin(Im(evals[1])*t);
  yg:=add(c[k]*Y[k], k=1..2);
```

$$Y_1 := \cos(4\,t)$$

$$Y_2 := \sin(4\,t)$$

$$yg := c_1\cos(4\,t) + c_2\sin(4\,t)$$

```
> eq1:=eval(subs(t=0, yg)=-1/2);
  eq2:=eval(subs(t=0, diff(yg, t))=2);
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 = \frac{-1}{2}$$

$$eq2 := 4\,c_2 = 2$$

$$val\_c := \{c_1 = \frac{-1}{2}, c_2 = \frac{1}{2}\}$$

```
> sol:=subs(val_c, yg);
```

$$sol := -\frac{1}{2}\cos(4\,t) + \frac{1}{2}\sin(4\,t)$$

Finally, bring the solution into phase-amplitude form.

```
> PA:=A*sin(4*t+phi);
  pars:=solve(identity(sol=PA, t), {A, phi});
```

$$PA := A\sin(4\,t + \phi)$$

$$pars := \{A = \frac{\sqrt{2}}{2}, \phi = -\frac{\pi}{4}\}, \{A = -\frac{\sqrt{2}}{2}, \phi = \frac{3\,\pi}{4}\}$$

```
> solPA:=subs(pars[1], PA);
```

$$solPA := \frac{1}{2}\sqrt{2}\,\sin\left(4\,t - \frac{\pi}{4}\right)$$

Clearly, the amplitude *A*, period *T*, and frequency *f*, are given by

$$A=\frac{\sqrt{2}}{2},\ T=\frac{\pi}{2} \quad \text{and} \quad f=\frac{2}{\pi}$$

The mass passes for the first time through the equilibrium position at the time that corresponds to the first positive zero of $\sin\left(4\,t-\frac{\pi}{4}\right)$. That zero is the solution of the equation $4\,t-\frac{\pi}{4}=0$, so $t=\frac{\pi}{16}$.

```
> A:=evalf(subs(pars[1], A));
  T:=evalf(Pi/2);
  f:=evalf(1/T);
```

$$A := 0.7071067810$$
$$T := 1.570796327$$
$$f := 0.6366197723$$

```
>
```

8: Page 219

Code the differential equation, find the auxiliary equation, compute the eigenvalues.

```
> deq:=20*diff(y(t), t$2)+140*diff(y(t), t)+200*y(t)=0;
  aux:=simplify(eval(subs(y(t)=exp(r*t), deq))/exp(r*t));
  evals:=solve(aux, r);
```

$$deq := 20\left(\frac{d^2}{dt^2}\,y(t)\right)+140\left(\frac{d}{dt}\,y(t)\right)+200\,y(t)=0$$
$$aux := 20\,r^2+140\,r+200=0$$
$$evals := -2, -5$$

Find the general solution and implement the initial conditions.

```
> for k to 2 do
  Y[k]:=exp(evals[k]*t);
  od;
  yg:=add(c[k]*Y[k], k=1..2);
```

$$Y_1 := e^{(-2\,t)}$$
$$Y_2 := e^{(-5\,t)}$$
$$yg := c_1\,e^{(-2\,t)}+c_2\,e^{(-5\,t)}$$

```
> eq1:=eval(subs(t=0, yg)=1/4);
  eq2:=eval(subs(t=0, diff(yg, t))=-1);
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1+c_2=\frac{1}{4}$$
$$eq2 := -2\,c_1-5\,c_2=-1$$
$$val\_c := \{c_2=\frac{1}{6},\ c_1=\frac{1}{12}\}$$

```
> sol:=subs(val_c, yg);
```

$$sol := \frac{1}{12}\,e^{(-2\,t)}+\frac{1}{6}\,e^{(-5\,t)}$$

Since this expression is always positive, the mass never returns to the equilibrium position.

```
>
```

11: Page 220

Code the differential equation, find the auxiliary equation, compute the eigenvalues.

```
> deq:=1*diff(y(t), t$2)+1/5*diff(y(t), t)+100*y(t)=0;
  aux:=simplify(eval(subs(y(t)=exp(r*t), deq))/exp(r*t));
  evals:=solve(aux, r);
```

$$deq := \left(\frac{d^2}{dt^2}\mathrm{y}(t)\right) + \frac{1}{5}\left(\frac{d}{dt}\mathrm{y}(t)\right) + 100\,\mathrm{y}(t) = 0$$

$$aux := r^2 + \frac{1}{5}r + 100 = 0$$

$$evals := -\frac{1}{10} + \frac{3}{10}I\sqrt{1111},\ -\frac{1}{10} - \frac{3}{10}I\sqrt{1111}$$

Find the general solution and implement the initial conditions.

```
> Y[1]:=exp(Re(evals[1])*t)*cos(Im(evals[1])*t);
  Y[2]:=exp(Re(evals[1])*t)*sin(Im(evals[1])*t);
  yg:=add(c[k]*Y[k], k=1..2);
```

$$Y_1 := \mathbf{e}^{\left(-\frac{t}{10}\right)}\cos\left(\frac{3\sqrt{1111}\,t}{10}\right)$$

$$Y_2 := \mathbf{e}^{\left(-\frac{t}{10}\right)}\sin\left(\frac{3\sqrt{1111}\,t}{10}\right)$$

$$yg := c_1\,\mathbf{e}^{\left(-\frac{t}{10}\right)}\cos\left(\frac{3\sqrt{1111}\,t}{10}\right) + c_2\,\mathbf{e}^{\left(-\frac{t}{10}\right)}\sin\left(\frac{3\sqrt{1111}\,t}{10}\right)$$

```
> eq1:=eval(subs(t=0, yg)=0);
  eq2:=eval(subs(t=0, diff(yg, t))=1);
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 = 0$$

$$eq2 := -\frac{1}{10}c_1 + \frac{3}{10}c_2\sqrt{1111} = 1$$

$$val\_c := \{c_1 = 0,\ c_2 = \frac{10\sqrt{1111}}{3333}\}$$

```
> sol:=subs(val_c, yg);
```

$$sol := \frac{10}{3333}\sqrt{1111}\,\mathbf{e}^{\left(-\frac{t}{10}\right)}\sin\left(\frac{3\sqrt{1111}\,t}{10}\right)$$

Plot a graph.

```
> plot(sol, t=0..1);
```



Compute the derivative of the solution and find a zero of that derivative between 0 and 0.2.

```
> der:=diff(sol, t);
  tmax:=fsolve(der=0, t=0..0.2);
```

$$der := -\frac{1}{3333}\sqrt{1111}\,\mathbf{e}^{\left(-\frac{t}{10}\right)}\sin\left(\frac{3\sqrt{1111}\,t}{10}\right) + \mathbf{e}^{\left(-\frac{t}{10}\right)}\cos\left(\frac{3\sqrt{1111}\,t}{10}\right)$$

$$tmax := 0.1560874206$$

Of course, with a little bit of effort, an exact value can be found too.

```
> _EnvAllSolutions:=true;
  t_extrema:=solve(der=0, t);
```

$$\_EnvAllSolutions := true$$

$$t\_extrema := \frac{10}{3333}(\arctan(3\sqrt{1111}) + \pi\,\_Z1)\sqrt{1111}$$

The smallest positive value is obtained when _Z1 = 0.

```
> tmax_exact:=subs(_Z1=0, t_extrema);
  evalf(tmax_exact);
```

$$tmax\_exact := \frac{10}{3333}\arctan(3\sqrt{1111})\sqrt{1111}$$

$$0.1560874206$$

```
>
```

16: Page 220

Setup the two differential equations and find the period of the general solutions. Be sure to clear the variable *k*, because it has been used as a do-loop counter in Exercise 8. Then, declare *m* and *k* to be positive.

```
> k:='k';
  interface(showassumed=0);
  assume(m>0);
  assume(k>0);
```

$$k := k$$

$$1$$

Part i. The original mass.

```
> deq1:=m*diff(y(t),t$2)+k*y(t)=0;
  aux1:=simplify(eval(subs(y(t)=exp(r*t), deq1))/exp(r*t));
  evals1:=solve(aux1, r);
```

$$deq1 := m\left(\frac{d^2}{dt^2}\,y(t)\right) + k\,y(t) = 0$$

$$aux1 := m\,r^2 + k = 0$$

$$evals1 := \frac{\sqrt{m\,k}\,I}{m}, \frac{-I\sqrt{m\,k}}{m}$$

Clearly, the period of this solution equals

```
> T1:=2*Pi/sqrt(k/m);
```

$$T1 := \frac{2\pi}{\sqrt{\frac{k}{m}}}$$

This gives us the first of two equations for *m* and *k*.

```
> eq1:=T1=3;
```

$$eq1 := \frac{2\pi}{\sqrt{\frac{k}{m}}} = 3$$

```
>
```

Part ii. The original mass + 2 kg.

```
> deq2:=(m+2)*diff(y(t),t$2)+k*y(t)=0;
  aux2:=simplify(eval(subs(y(t)=exp(r*t), deq2))/exp(r*t));
  evals2:=solve(aux2, r);
```

$$deq2 := (m+2)\left(\frac{d^2}{dt^2}\,y(t)\right) + k\,y(t) = 0$$

$$aux2 := m\,r^2 + 2\,r^2 + k = 0$$

$$evals2 := \frac{\sqrt{(m+2)\,k}\,I}{m+2}, \frac{-I\sqrt{(m+2)\,k}}{m+2}$$

Clearly, the period of this solution equals

```
> T2:=2*Pi/sqrt(k/(m+2));
```

$$T2 := \frac{2\pi}{\sqrt{\frac{k}{m+2}}}$$

This gives us the second of two equations for *m* and *k*.

```
> eq2:=T2=4;
```

$$eq2 := \frac{2\pi}{\sqrt{\frac{k}{m+2}}} = 4$$

```
>
```

Part iii. The solution of the equations for *m* and *k*.

```
> pars:=solve({eq1, eq2}, {m,k});
  parsf:=evalf(pars);
```

$$pars := \left\{k = \frac{8\pi^2}{7}, m = \frac{18}{7}\right\}$$

$$parsf := \{m = 2.571428571, k = 11.27954789\}$$

We conclude that the original mass was $\frac{18}{7} = 2.571428571$ kg.

```
>
```